NASA
Technical
Memorandum

NASA TM- 103509

TRANSTRAIN: A PROGRAM TO COMPUTE STRAIN TRANSFORMATIONS IN COMPOSITE MATERIALS

By Rafiq Ahmed

Structures and Dynamics Laboratory
Science and Engineering Directorate

July 1990

National Aeronautics and
Space Administration
George C. Marshall Space Flight Center

NASA
National Aeronautics and Space Administration

Report Documentation Page

1. Report No. NASA TM - 103509	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle TRANSTRAIN: A Program to Compute Strain Transformations in Composite Materials		5. Report Date July 1990
		6. Performing Organization Code
7. Author(s) Rafiq Ahmed		8. Performing Organization Report No.
		10. Work Unit No.
9. Performing Organization Name and Address George C. Marshall Space Flight Center Marshall Space Flight Center, Alabama 35812		11. Contract or Grant No.
		13. Type of Report and Period Covered Technical Memorandum
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		14. Sponsoring Agency Code

15. Supplementary Notes

Prepared by Structures and Dynamics Laboratory, Science and Engineering Directorate.

16. Abstract

This report documents a computer program that generates strain transformations and on-axis stresses in composites given the initial strains and the transformation angles.

17. Key Words (Suggested by Author(s)) composite materials, carbon phenolic, carbon-carbon, coordinate transformation, FORTRAN	18. Distribution Statement Unclassified – Unlimited

19. Security Classif. (of this report) Unclassified	20. Security Classif. (of this page) Unclassified	21. No. of pages 17	22. Price NTIS

NASA FORM 1626 OCT 86

TABLE OF CONTENTS

TECHNICAL MEMORANDUM

TRANSTRAIN: A PROGRAM TO COMPUTE STRAIN TRANSFORMATIONS IN COMPOSITE MATERIALS

INTRODUCTION

Over the years, the solid rocket motor community has made increasing use of composite materials for thermal and structural applications. This is particularly true of solid rocket nozzles, which have used carbon phenolic and, increasingly, carbon-carbon materials to provide structural integrity and thermal protection at the high temperatures encountered during motor burn.

To evaluate the degree of structural performance of nozzles and their materials and to verify analysis models, many subscale and full-scale tests are run. These provide engineers with valuable data needed to optimize design and to analyze nozzle hardware. Included among these data are strains, pressures, thrust, temperatures, and displacements. Recent nozzle test hardware has made increasing use of strain gauges embedded in the carbon composite material to measure internal strains. In order to evaluate strength, these data must be transformed into strains along the fiber directions. The fiber-direction stresses can then be calculated. This report concentrates on a computer program written to help engineers correctly manipulate the strain data into a form that can be used to evaluate structural integrity of the nozzle.

OBJECTIVE

The objective of this report is to describe a tool that can be used to transform strains from an arbitrary direction to the on-axis strains along and orthogonal to the fiber directions. The on-axis stresses are subsequently calculated. A computer program was developed to simplify and ease this process.

THEORETICAL BACKGROUND

The program performs the following two steps to find the desired data:

1. Three-axis strain rotation to find the transformed strains

2. On-axis stress computation for an orthotropic material using Gaussian elimination given the compliance matrix.

The program begins with the input data read from the file STR.INP, which contains the three input transformation angles (THETA, GAMMA, and ALPHA), the material properties (Young's moduli, Poisson's ratios, and shear moduli), and the input strains. An orthotropic material is assumed (i.e., no coupling exists between normal and shear stresses).

The three-axis rotation was chosen because it is the most general case and will cover most composite orientations. The initial coordinate system is defined as $\{x_1,y_1,z_1\}$. The second coordinate system (after the first transformation) is defined as $\{x_2,y_2,z_2\}$. The third coordinate system is defined as $\{x_3,y_3,z_3\}$ with the final, transformed coordinate system being defined as $\{x_4,y_4,z_4\}$. θ is the angle of rotation about the z_1 axis, with γ and α being the respective rotations about the y_2 and x_3 axes. A typical coordinate transformation is shown diagramatically in figure 1. The three coordinate transformations are, respectively:

$$\begin{Bmatrix} x_2 \\ y_2 \\ z_2 \end{Bmatrix} = \begin{bmatrix} \cos\theta & \sin\theta & 0 \\ -\sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{Bmatrix} x_1 \\ y_1 \\ z_1 \end{Bmatrix}$$

$$\begin{Bmatrix} x_3 \\ y_3 \\ z_3 \end{Bmatrix} = \begin{bmatrix} \cos\gamma & 0 & \sin\gamma \\ 0 & 1 & 0 \\ -\sin\gamma & 0 & \cos\gamma \end{bmatrix} \begin{Bmatrix} x_2 \\ y_2 \\ z_2 \end{Bmatrix}$$

$$\begin{Bmatrix} x_4 \\ y_4 \\ z_4 \end{Bmatrix} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & \sin\alpha \\ 0 & -\sin\alpha & \cos\alpha \end{bmatrix} \begin{Bmatrix} x_3 \\ y_3 \\ z_3 \end{Bmatrix}.$$

The combined transformation matrix, $[a]$, is found by matrix multiplication:

$$\begin{bmatrix} \cos\gamma\cos\theta & \cos\gamma\sin\theta & \sin\gamma \\ (-\sin\alpha\sin\gamma\cos\theta - \cos\alpha\sin\theta) & (-\sin\alpha\sin\gamma\sin\theta + \cos\alpha\cos\theta) & \sin\alpha\cos\gamma \\ (-\cos\alpha\sin\gamma\cos\theta + \sin\alpha\sin\theta) & (-\cos\alpha\sin\gamma\sin\theta - \sin\alpha\cos\theta) & \cos\alpha\cos\gamma \end{bmatrix}$$

After this matrix is calculated, the transformed strains are calculated by using the continuum mechanics relation for strain transformations:

$$\varepsilon_{ij} = a_{ir}\, a_{js}\, \varepsilon_{rs}$$

After the transformed strains are calculated, the subroutine TRANS converts them from double index notation to siingle index notations as follows:

$$\varepsilon_1 = \varepsilon_{11}$$

$$\varepsilon_2 = \varepsilon_{22}$$

$$\varepsilon_3 = \varepsilon_{33}$$

$$\varepsilon_4 = \varepsilon_{23}$$

$$\varepsilon_5 = \varepsilon_{13}$$

$$\varepsilon_6 = \varepsilon_{12}$$

Finally, the subroutines ORTHOCOMP and ORTHOSTR are called to calculate the on-axis stresses. This is accomplished by solving for the stresses from the strain-stress relationship using Gaussian elimination (performed by GAUSS). The strain-stress relationship (compliance matrix) is derived from the material properties in the subroutine ORTHOCOMP. The strain-stress relationship is shown in the following expression:

$$\begin{Bmatrix} \varepsilon_1 \\ \varepsilon_2 \\ \varepsilon_3 \\ \varepsilon_4 \\ \varepsilon_5 \\ \varepsilon_6 \end{Bmatrix} = \begin{bmatrix} 1/E_1 & -\nu_{12}/E_2 & -\nu_{13}/E_3 & 0 & 0 & 0 \\ -\nu_{21}/E_1 & 1/E_2 & -\nu_{23}/E_3 & 0 & 0 & 0 \\ -\nu_{31}/E_1 & -\nu_{32}/E_2 & 1/E_3 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1/G_{23} & 0 & 0 \\ 0 & 0 & 0 & 0 & 1/G_{13} & 0 \\ 0 & 0 & 0 & 0 & 0 & 1/G_{12} \end{bmatrix} \begin{Bmatrix} \sigma_1 \\ \sigma_2 \\ \sigma_3 \\ \sigma_4 \\ \sigma_5 \\ \sigma_6 \end{Bmatrix}$$

After all these calculations, the output is written to an output file, "TRSTRAIN.OUT."

TRANSTRAIN USER'S GUIDE

The TRANSTRAIN program is currently resident on the Marshall Space Flight Center's IESL2VAX4 computer under the name "TRANSTRAIN.FOR" and can be copied from the subdirectory ZFA3:[024030.FORTRAN] TRANSTRAIN.FOR. A FORTRAN listing for the program along with sample input and output files are given in the appendix. The input file is written in free format (with commas between each of the entries on a given line) and must be given the name "STR.INP."

The first line should contain the following items in order:

THETA, GAMMA, ALPHA, E1,E2,E3

where THETA is the rotation angle about the z_1 axis, GAMMA is the rotation angle about the y_2 axis, ALPHA is the rotation angle about the x_3 axis, E1 is the warp direction modulus of elasticity, E2 is the fill direction modulus of elasticity, and E3 is the normal direction (at right angles to both the warp and fill directions) modulus of elasticity. The user should keep in mind that THETA, GAMMA, and ALPHA should be chosen in a coordinate system consistent with the initial strain orientations.

The second line of the input file should contain the following items in order:

NU21, NU32, NU31, G12, G23, G13

where NU21 is the longitudinal fill-warp Poisson's ratio, NU32 is the longitudinal normal-fill Poisson's ratio, NU31 is the longitudinal normal-warp Poisson's ratio, G12 is the warp-fill shear modulus, G23 is the fill-normal shear modulus, and G13 is the warp-normal shear modulus.

Each subsequent line should contain one input strain in the following order:

EPS11,EPS12,EPS13,EPS21,EPS22,EPS23,EPS31,EPS32,EPS33

Each of these lines should contain only one strain. The user should note that the following must be true of the input strain matrix: EPS12 = EPS21, EPS13 = EPS31, and EPS23 = EPS32.

The input file is now complete. To run the program TRANSTRAIN, the user should follow the FORTRAN run/link procedure prescribed for the computer he is using. The output file is named "TRSTRAIN.OUT" and is created automatically by the program.

SUMMARY

This code is a relatively simple, easy to use tool for computing three-axis strain transformations and the resulting stresses for an orthotropic material. Since the code was written by making extensive use of subroutines, it is possible to add additional subroutines to take into account monoclinic materials or other material models where shear coupling occurs. This program can thus prove very useful in both testing and analysis applications.



Figure 1. Typical coordinate transformation.

APPENDIX

```
C  ****************************************
C PROGRAM TRANSTRAIN
C This program computes the transformed strains from a 3-axis
C rotation and calculates the compliance matrix.  The on-axis
C stresses are then calculated.
C  ****************************************
C
      PROGRAM TRANSTRAIN
      DIMENSION EPSP(3,3), EPS(3,3), EP(6), S(6,6), SIGON(6)
      REAL THETA,GAMMA,ALPHA,T,E1,E2,E3,NU21,NU32,NU31,G12,G23,G13
      INTEGER N
      N=6
C
C Zero out variables and arrays
C
      THETA=0.0
      GAMMA=0.0
      ALPHA=0.0
      DO 2001 I=1,3
        DO 2002 J=1,3
                EPSP(I,J)=0.0
                EPS(I,J)=0.0
 2002 CONTINUE
 2001 CONTINUE
      DO 3001 I=1,6
        EP(I)=0.0
        SIGON(I)=0.0
 3001 CONTINUE
      DO 4001 I=1,6
          DO 4002 J=1,6
             S(I,J)=0.0
 4002 CONTINUE
 4001 CONTINUE
C
C Open input and output files
C
      OPEN (UNIT=14,FILE='STR.INP',STATUS='OLD')
      OPEN (UNIT=16,FILE='TRSTRAIN.OUT',STATUS='NEW')
      DO 1001 L=1,3
           DO 1002 M=1,3
                EPSP(L,M)=0
                EPS(L,M)=0
 1002 CONTINUE
 1001 CONTINUE
C
C Read and write out input data
C
      WRITE(16,101)
  101 FORMAT(5X,'*********************************************')
      WRITE(16,102)
  102 FORMAT(5X,'TRANSFORMED STRAIN AND ON-AXIS STRESS OUTPUT')
      WRITE(16,103)
  103 FORMAT(5X,'*********************************************')
      READ(14,*) THETA,GAMMA,ALPHA,E1,E2,E3
      READ(14,*) NU21,NU32,NU31,G12,G23,G13
      WRITE(16,120)
  120 FORMAT(5X,'THETA',5X,'GAMMA',5X,'ALPHA')
      WRITE(16,121) THETA,GAMMA,ALPHA
  121 FORMAT(6X,F4.1,6X,F4.1,6X,F4.1)
      WRITE(16,122)
  122 FORMAT(/,5X,'E1',9X,'E2',9X,'E3')
      WRITE(16,123) E1,E2,E3
  123 FORMAT(5X,F9.1,2X,F9.1,2X,F9.1)
      WRITE(16,124)
```

```
  124 FORMAT(/,5X,'NU21',5X,'NU32',5X,'NU31')
      WRITE(16,125) NU21,NU32,NU31
  125 FORMAT(5X,F4.3,5X,F4.3,5X,F4.3)
      WRITE(16,126)
  126 FORMAT(/,5X,'G12',8X,'G23',8X,'G13')
      WRITE(16,127) G12,G23,G13
  127 FORMAT(5X,F9.1,2X,F9.1,2X,F9.1)
      WRITE(16,104)
  104 FORMAT(/,5X,'INPUT STRAINS')
      DO 98 I=1,3
            DO 99 J=1,3
                  READ(14,*) EPSP(I,J)
   99 CONTINUE
   98 CONTINUE
      DO 80 I=1,3
            DO 81 J=1,3
                  WRITE(16,105) I,J,EPSP(I,J)
   81 CONTINUE
   80 CONTINUE
  105 FORMAT(5X,'EPSP(',I1,',',I1,') = ',F10.7)
C
C Call subroutines to calculate transformed strains
C
      CALL TRSTRAIN(THETA,GAMMA,ALPHA,EPSP,EPS)
      CALL TRANS(EPS,EP)

C
C Write out transformed strains
C
      WRITE(16,107)
  107 FORMAT(/,5X,'TRANSFORMED STRAINS')
      DO 70 I=1,6
           WRITE(16,106) I,EP(I)
   70 CONTINUE
  106 FORMAT(5X,'EP(',I1,') = ',F10.7)

C
C Call subroutine to calculate compliance matrix
C
      CALL ORTHOCOMP(EP,S,E1,E2,E3,NU21,NU32,NU31,G12,G23,G13)
C
C Write out S matrix
C
           WRITE(16,108)
  108 FORMAT(//,5X,'S MATRIX',/)
      DO 50 I=1,6
               WRITE(16,109) (S(I,J),J=1,6)
  109 FORMAT(5X,E10.4,2X,E10.4,2X,E10.4,2X,E10.4,2X,E10.4,2X,E10.4)
   50 CONTINUE
C
C Call subroutine to calculate on-axis stresses
C
      CALL ORTHOSTR(EP,S,SIGON,N)
C
C Write out on-axis stresses
C
           WRITE(16,110)
  110 FORMAT(//,5X,'ON-AXIS STRESSES')
      DO 40 I=1,6
           WRITE(16,111) I,SIGON(I)
   40 CONTINUE
  111 FORMAT(5X,'SIG(',I1,') = ',F10.5)
      STOP
      END
```

```
C
C ******************************************************
C This subroutine computes the transformed strains given
C input strain values
C ******************************************************
C
      SUBROUTINE TRSTRAIN(TH,GAM,ALP,EPSP,EPS)
      DIMENSION EPSP(3,3), EPS(3,3), A(3,3)
C
C Calculate strain transformation matrix
C
      PI = 3.141592638
      A(1,1) = (COS(GAM*PI/180))*(COS(TH*PI/180))
      A(1,2) = (COS(GAM*PI/180))*(SIN(TH*PI/180))
      A(1,3) = SIN(GAM*PI/180)
      A(2,1) = -(SIN(ALP*PI/180))*(SIN(GAM*PI/180))*(COS(TH*PI/180))
     &-(COS(ALP*PI/180))*(SIN(TH*PI/180))
      A(2,2) = -(SIN(ALP*PI/180))*(SIN(GAM*PI/180))*(SIN(TH*PI/180))
     &+(COS(ALP*PI/180))*(COS(TH*PI/180))
      A(2,3) = (SIN(ALP*PI/180))*(COS(GAM*PI/180))
      A(3,1) = -(COS(ALP*PI/180))*(SIN(GAM*PI/180))*(COS(TH*PI/180))
     &+(SIN(ALP*PI/180))*(SIN(TH*PI/180))
      A(3,2) = -(COS(ALP*PI/180))*(SIN(GAM*PI/180))*(SIN(TH*PI/180))
     &-(SIN(ALP*PI/180))*(COS(TH*PI/180))
      A(3,3) = (COS(ALP*PI/180))*(COS(GAM*PI/180))
C
C Use strain transformation relation to find transformed strains
C
      DO 1005 I = 1,3
         DO 1006 J = 1,3
            EPS(I,J) = 0
              DO 1007 R = 1,3
                DO 1008 S = 1,3
                  T = A(I,R)*A(J,S)*EPSP(R,S)
                  EPS(I,J) = EPS(I,J) + T
 1008 CONTINUE
 1007 CONTINUE
 1006 CONTINUE
 1005 CONTINUE
      RETURN
      END
C
C
C **********************************************************
C This subroutine computes the transformed strains in single
C index notation
C **********************************************************
C
        SUBROUTINE TRANS(EPS,EP)
        DIMENSION EPS(3,3), EP(6)
        EP(1) = EPS(1,1)
        EP(2) = EPS(2,2)
        EP(3) = EPS(3,3)
        EP(4) = EPS(2,3)
        EP(5) = EPS(1,3)
        EP(6) = EPS(1,2)
        RETURN
        END
C
C
C ***************************************************************
C This subroutine calculates the compliance matrix for an ortho-
C tropic material
C ***************************************************************
```

```
C
        SUBROUTINE ORTHOCOMP(EP,S,E1,E2,E3,NU21,NU32,
     &  NU31,G12,G23,G13)
        DIMENSION EP(6), S(6,6)
        REAL E1,E2,E3,NU12,NU21,NU23,NU32,NU13,NU31,G12,G23,G13,N,N1
        NU12 = NU21*E2/E1
        NU23 = NU32*E3/E2
        NU13 = NU31*E3/E1
        DO 1001 I=1,6
          DO 1002 J=1,6
               S(I,J)=0
 1002   CONTINUE
 1001   CONTINUE
C
C Compute each term of the S (compliance) matrix for orthotropic material
C
        S(1,1) = 1/E1
        S(1,2) = -NU12/E2
        S(1,3) = -NU13/E3
        S(2,1) = -NU21/E1
        S(2,2) = 1/E2
        S(2,3) = -NU23/E3
        S(3,1) = -NU31/E1
        S(3,2) = -NU32/E2
        S(3,3) = 1/E3
        S(4,4) = 1/G23
        S(5,5) = 1/G13
        S(6,6) = 1/G12
        RETURN
        END
C
C *****************************************************************
C This subroutine calculates the on-axis stresses from the compliance
C matrix and the transformed strains
C *****************************************************************
        SUBROUTINE ORTHOSTR(EP,S,SIGON,N)
        DIMENSION EP(6), S(6,6), SIGON(6)
        INTEGER N
C
C Call subroutine GAUSS to solve linear equations
C
        CALL GAUSS(S,EP,N)
        DO 500 I=1,N
            SIGON(I) = EP(I)
  500   CONTINUE
        RETURN
        END
C
C ***********************************************************************
C This subroutine solves a system of linear equations via the Gauss-Jordan
C method of elimination
C ***********************************************************************
C
        SUBROUTINE GAUSS(A,B,N)
        DIMENSION A(6,6),B(6),C(6)
        REAL SUM
        INTEGER N,N1,K1,K
        N1=N-1
        DO 100 K=1,N1
             K1=K+1
             DO 200 L=K,N
                 C(L) = A(K,L)
  200        CONTINUE
             AKK = 1/C(K)
```

```
            BK = B(K)
            DO 300 I=K1,N
                AIK = A(I,K)*AKK
                B(I) = B(I) - AIK*BK
                DO 400 J=K,N
                    A(I,J) = A(I,J) - AIK*C(J)
400             CONTINUE
300         CONTINUE
100     CONTINUE
        K=N
        B(K) = B(K)/A(K,K)
470     K=K-1
        IF (K .LE. 0) THEN
            GOTO 560
        END IF
        K1=K+1
        SUM=0
        DO 700 J=K1,N
            SUM = SUM + A(K,J)*B(J)
700     CONTINUE
        B(K) = (B(K) - SUM)/A(K,K)
        GOTO 470
560     RETURN
        END
```

```
30.,30.,0.,2600000.,2600000.,2360000.
0.32,0.232,0.232,1000000.,1000000.,1000000.
0.001
0.000
0.000
0.000
0.000
0.000
0.000
0.000
0.000
```

This is the STR.INP file

```
*********************************************
TRANSFORMED STRAIN AND ON-AXIS STRESS OUTPUT
*********************************************
THETA      GAMMA      ALPHA
 30.0       30.0         .0

E1          E2          E3
2600000.0  2600000.0  2360000.0

NU21      NU32      NU31
.320      .232      .232

G12         G23         G13
1000000.0  1000000.0  1000000.0

INPUT STRAINS
EPSP(1,1) =   .0010000
EPSP(1,2) =   .0000000
EPSP(1,3) =   .0000000
EPSP(2,1) =   .0000000
EPSP(2,2) =   .0000000
EPSP(2,3) =   .0000000
EPSP(3,1) =   .0000000
EPSP(3,2) =   .0000000
EPSP(3,3) =   .0000000

TRANSFORMED STRAINS
EP(1) =   .0005625
EP(2) =   .0002500
EP(3) =   .0001875
EP(4) =   .0002165
EP(5) =  -.0003248
EP(6) =  -.0003750


S MATRIX

0.3846E-06  -.1231E-06  -.8923E-07  0.0000E+00  0.0000E+00  0.0000E+00
-.1231E-06  0.3846E-06  -.8923E-07  0.0000E+00  0.0000E+00  0.0000E+00
-.8923E-07  -.8923E-07  0.4237E-06  0.0000E+00  0.0000E+00  0.0000E+00
0.0000E+00  0.0000E+00  0.0000E+00  0.1000E-05  0.0000E+00  0.0000E+00
0.0000E+00  0.0000E+00  0.0000E+00  0.0000E+00  0.1000E-05  0.0000E+00
0.0000E+00  0.0000E+00  0.0000E+00  0.0000E+00  0.0000E+00  0.1000E-05


ON-AXIS STRESSES
SIG(1) = 2298.03220
SIG(2) = 1682.50200
SIG(3) = 1280.73920
SIG(4) =  216.50631
SIG(5) = -324.75952
SIG(6) = -375.00000
```

BIBLIOGRAPHY

Tsai, S.W., and Hahn, H.T.: "Introduction to Composite Materials." Technomic, 1980.

Tsai, S.W.: "Composites Design." Think Composites, 1988.

Chung, T.J.: "Continuum Mechanics." McGraw-Hill, 1988.

Ellis, T.M.R.: "A Structured Approach to FORTRAN 77 Programming." Addison-Wesley, 1982.

APPROVAL

TRANSTRAIN: A PROGRAM TO COMPUTE STRAIN TRANSFORMATIONS IN COMPOSITE MATERIALS

By Rafiq Ahmed

The information in this report has been reviewed for technical content. Review of any information concerning Department of Defense or nuclear energy activities or programs has been made by the MSFC Security Classification Officer. This report, in its entirety, has been determined to be unclassified.

J. Blair

JAMES C. BLAIR
Director, Structures and Dynamics Laboratory

☆U.S. GOVERNMENT PRINTING OFFICE 1990–731–061/20203